Rules for integrands of the form $u\ (a + b\ \text{ArcTanh}[c + d\ x])^p$

1. $\int (a + b\ \text{ArcTanh}[c + d\ x])^p\, dx$

1: $\int (a + b\ \text{ArcTanh}[c + d\ x])^p\, dx$ when $p \in \mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Rule: If $p \in \mathbb{Z}^+$, then**

$$\int (a + b\ \text{ArcTanh}[c + d\ x])^p\, dx \longrightarrow \frac{1}{d}\ \text{Subst}\left[\int (a + b\ \text{ArcTanh}[x])^p\, dx,\ x,\ c + d\ x\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcTanh[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcTanh[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0]
```

```
Int[(a_.+b_.*ArcCoth[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcCoth[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0]
```

U: $\int (a + b\ \text{ArcTanh}[c + d\ x])^p\, dx$ when $p \notin \mathbb{Z}^+$

- **Rule: If $p \notin \mathbb{Z}^+$, then**

$$\int (a + b\ \text{ArcTanh}[c + d\ x])^p\, dx \longrightarrow \int (a + b\ \text{ArcTanh}[c + d\ x])^p\, dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcTanh[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcTanh[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,p},x] && Not[IGtQ[p,0]]
```

```
Int[(a_.+b_.*ArcCoth[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcCoth[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,p},x] && Not[IGtQ[p,0]]
```

2. $\int (e + f x)^m (a + b \operatorname{ArcTanh}[c + d x])^p \, dx$

1: $\int (e + f x)^m (a + b \operatorname{ArcTanh}[c + d x])^p \, dx$ when $d e - c f == 0 \land p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $d e - c f == 0 \land p \in \mathbb{Z}^+$**, then**

$$\int (e + f x)^m (a + b \operatorname{ArcTanh}[c + d x])^p \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(\frac{f x}{d}\right)^m (a + b \operatorname{ArcTanh}[x])^p \, dx,\ x,\ c + d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcTanh[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(f*x/d)^m*(a+b*ArcTanh[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[d*e-c*f,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCoth[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(f*x/d)^m*(a+b*ArcCoth[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[d*e-c*f,0] && IGtQ[p,0]
```

2: $\int (e + f x)^m (a + b \operatorname{ArcTanh}[c + d x])^p \, dx$ when $p \in \mathbb{Z}^+ \land m + 1 \in \mathbb{Z}^-$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{ArcTanh}[c + d x])^p == \frac{b d p (a+b \operatorname{ArcTanh}[c+d x])^{p-1}}{1-(c+d x)^2}$

Rule: If $p \in \mathbb{Z}^+ \land m + 1 \in \mathbb{Z}^-$**, then**

$$\int (e + f x)^m (a + b \operatorname{ArcTanh}[c + d x])^p \, dx \longrightarrow \frac{(e + f x)^{m+1} (a + b \operatorname{ArcTanh}[c + d x])^p}{f (m + 1)} - \frac{b d p}{f (m + 1)} \int \frac{(e + f x)^{m+1} (a + b \operatorname{ArcTanh}[c + d x])^{p-1}}{1 - (c + d x)^2} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_*(a_.+b_.*ArcTanh[c_+d_.*x_])^p_.,x_Symbol] :=
  (e+f*x)^(m+1)*(a+b*ArcTanh[c+d*x])^p/(f*(m+1)) -
  b*d*p/(f*(m+1))*Int[(e+f*x)^(m+1)*(a+b*ArcTanh[c+d*x])^(p-1)/(1-(c+d*x)^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && ILtQ[m,-1]
```

```
Int[(e_.+f_.*x_)^m_*(a_.+b_.*ArcCoth[c_+d_.*x_])^p_.,x_Symbol] :=
  (e+f*x)^(m+1)*(a+b*ArcCoth[c+d*x])^p/(f*(m+1)) -
  b*d*p/(f*(m+1))*Int[(e+f*x)^(m+1)*(a+b*ArcCoth[c+d*x])^(p-1)/(1-(c+d*x)^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && ILtQ[m,-1]
```

3: $\int (e+f x)^m (a+b \operatorname{ArcTanh}[c+d x])^p \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (e+f x)^m (a+b \operatorname{ArcTanh}[c+d x])^p \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(\frac{d e-c f}{d}+\frac{f x}{d}\right)^m (a+b \operatorname{ArcTanh}[x])^p \, dx,\ x,\ c+d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcTanh[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcTanh[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCoth[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcCoth[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0]
```

U: $\int (e+f x)^m (a+b \operatorname{ArcTanh}[c+d x])^p \, dx$ when $p \notin \mathbb{Z}^+$

Rule: If $p \notin \mathbb{Z}^+$, then

$$\int (e+f x)^m (a+b \operatorname{ArcTanh}[c+d x])^p \, dx \longrightarrow \int (e+f x)^m (a+b \operatorname{ArcTanh}[c+d x])^p \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcTanh[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(e+f*x)^m*(a+b*ArcTanh[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IGtQ[p,0]]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCoth[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(e+f*x)^m*(a+b*ArcCoth[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IGtQ[p,0]]
```

3. $\int (e + f\,x^n)^m\,(a + b\,\text{ArcTanh}[c + d\,x])^p\,dx$

5. $\int \frac{\text{ArcTanh}[c + d\,x]}{e + f\,x^n}\,dx$

1: $\int \frac{\text{ArcTanh}[c + d\,x]}{e + f\,x^n}\,dx$ when $n \in \mathbb{Q}$

Derivation: Algebraic expansion

Basis: $\text{ArcTanh}[z] == \frac{1}{2}\,\text{Log}[1 + z] - \frac{1}{2}\,\text{Log}[1 - z]$

Basis: $\text{ArcCoth}[z] == \frac{1}{2}\,\text{Log}\left[\frac{1+z}{z}\right] - \frac{1}{2}\,\text{Log}\left[\frac{-1+z}{z}\right]$

Rule: If $n \in \mathbb{Q}$**, then**

$$\int \frac{\text{ArcTanh}[c + d\,x]}{e + f\,x^n}\,dx \longrightarrow \frac{1}{2}\int \frac{\text{Log}[1 + c + d\,x]}{e + f\,x^n}\,dx - \frac{1}{2}\int \frac{\text{Log}[1 - c - d\,x]}{e + f\,x^n}\,dx$$

Program code:

```
Int[ArcTanh[c_+d_.*x_]/(e_+f_.*x_^n_.),x_Symbol] :=
  1/2*Int[Log[1+c+d*x]/(e+f*x^n),x] -
  1/2*Int[Log[1-c-d*x]/(e+f*x^n),x] /;
FreeQ[{c,d,e,f},x] && RationalQ[n]
```

```
Int[ArcCoth[c_+d_.*x_]/(e_+f_.*x_^n_.),x_Symbol] :=
  1/2*Int[Log[(1+c+d*x)/(c+d*x)]/(e+f*x^n),x] -
  1/2*Int[Log[(-1+c+d*x)/(c+d*x)]/(e+f*x^n),x] /;
FreeQ[{c,d,e,f},x] && RationalQ[n]
```

U: $\int \frac{\text{ArcTanh}[c+d\,x]}{e+f\,x^n}\,dx$ when $n \notin \mathbb{Q}$

Rule: If $n \notin \mathbb{Q}$**, then**

$$\int \frac{\text{ArcTanh}[c+d\,x]}{e+f\,x^n}\,dx \longrightarrow \int \frac{\text{ArcTanh}[c+d\,x]}{e+f\,x^n}\,dx$$

Program code:

```
Int[ArcTanh[c_+d_.*x_]/(e_+f_.*x_^n_),x_Symbol] :=
  Unintegrable[ArcTanh[c+d*x]/(e+f*x^n),x] /;
FreeQ[{c,d,e,f,n},x] && Not[RationalQ[n]]
```

```
Int[ArcCoth[c_+d_.*x_]/(e_+f_.*x_^n_),x_Symbol] :=
  Unintegrable[ArcCoth[c+d*x]/(e+f*x^n),x] /;
FreeQ[{c,d,e,f,n},x] && Not[RationalQ[n]]
```

4: $\int \left(A+B\,x+C\,x^2\right)^q (a+b\,\text{ArcTanh}[c+d\,x])^p\,dx$ when $B\left(1-c^2\right)+2\,A\,c\,d==0 \wedge 2\,c\,C-B\,d==0$

Derivation: Integration by substitution

- **Basis: If** $B\left(1-c^2\right)+2\,A\,c\,d==0 \wedge 2\,c\,C-B\,d==0$**, then** $A+B\,x+C\,x^2 == -\frac{C}{d^2}+\frac{C}{d^2}\,(c+d\,x)^2$
- **Rule: If** $B\left(1-c^2\right)+2\,A\,c\,d==0 \wedge 2\,c\,C-B\,d==0$**, then**

$$\int \left(A+B\,x+C\,x^2\right)^q (a+b\,\text{ArcTanh}[c+d\,x])^p\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int \left(-\frac{C}{d^2}+\frac{C\,x^2}{d^2}\right)^q (a+b\,\text{ArcTanh}[x])^p\,dx,\ x,\ c+d\,x\right]$$

- **Program code:**

```
Int[(A_.+B_.*x_+C_.*x_^2)^q_.*(a_.+b_.*ArcTanh[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(-C/d^2+C/d^2*x^2)^q*(a+b*ArcTanh[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,A,B,C,p,q},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

```
Int[(A_.+B_.*x_+C_.*x_^2)^q_.*(a_.+b_.*ArcCoth[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(C/d^2+C/d^2*x^2)^q*(a+b*ArcCoth[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,A,B,C,p,q},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

5: $\int (e+f x)^m \left(A+B x+C x^2\right)^q (a+b \operatorname{ArcTanh}[c+d x])^p \, dx$ when $B\left(1-c^2\right)+2 A c d == 0 \wedge 2 c C-B d == 0$

- **Derivation: Integration by substitution**
- **Basis: If** $B\left(1-c^2\right)+2 A c d == 0 \wedge 2 c C-B d == 0$**, then** $A+B x+C x^2 == -\frac{C}{d^2}+\frac{C}{d^2}(c+d x)^2$
- **Rule: If** $B\left(1-c^2\right)+2 A c d == 0 \wedge 2 c C-B d == 0$**, then**

$$\int (e+f x)^m \left(A+B x+C x^2\right)^q (a+b \operatorname{ArcTanh}[c+d x])^p \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(\frac{d e-c f}{d}+\frac{f x}{d}\right)^m \left(-\frac{C}{d^2}+\frac{C x^2}{d^2}\right)^q (a+b \operatorname{ArcTanh}[x])^p \, dx,\ x,\ c+d x\right]$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(A_.+B_.*x_+C_.*x_^2)^q_.*(a_.+b_.*ArcTanh[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(-C/d^2+C/d^2*x^2)^q*(a+b*ArcTanh[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,p,q},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

```
Int[(e_.+f_.*x_)^m_.*(A_.+B_.*x_+C_.*x_^2)^q_.*(a_.+b_.*ArcCoth[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(-C/d^2+C/d^2*x^2)^q*(a+b*ArcCoth[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,p,q},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```